
1 Overview

This document lists all the changes needed to correct issues in the ATS iwd-TTCN3-B2019-06_D20wk38 related to the title of this CR.

Contact:
Shaun Harry

shaun.harry@keysight.com
2 Corrections required
2.1 Change 1
	Function name
	Template cs_NR_SS_SpCellConfig

	Reason for change
	The template cs_NR_SS_SpCellConfig sets the field “MAC_CellGroupConfig.R15” and “PhysicalCellGroupConfig.R15” to the value of input paramters defined as “template (omit)” and can be passed as “omit”.
The field physicalCellGroupConfig in NRSpCellConfig_Type is defined as type NR_ASN1_PhysicalCellGroupConfig_Type, which is defined as :

 

type union NR_ASN1_PhysicalCellGroupConfig_Type {

    PhysicalCellGroupConfig R15

  }

 

Here “R15” is a union element, not an optional field so it cannot legally be set to “omit”. If “R15” is set to omit the resulting value is not fully defined and can give a run-time error if a “valueof” operation is performed. Similarly for type NR_ASN1_MAC_CellGroupConfig_Type.

Note: The issue has been exposed following the implementation of MCC commetns to R5s201322 which adds “valueof(p_NR_SYSTEM_CTRL_REQ.Request))” in function f_NR_SS_CommonCellConfig(). This gives a run-time error for cases where      MAC_CellGroupConfig or PhysicalCellGroupConfig have been set to “omit” in SpCellConfig.


	Summary of change
	Change the template paramters to use types MAC_CellGroupConfig and PhysicalCellGroupConfig, instead of NR_ASN1_MAC_CellGroupConfig_Type and NR_ASN1_PhysicalCellGroupConfig_Type.

Note : This allows “omit” to be passed into the template. For cases where values are passed for this paramters, “R15” needs to be set in the calling function or template.

	TTCN module
	NR_CellCfg_Templates.ttcn

	MCC160 Comment
	Accepted – implemented differently using local functions inside the template - see proposed implementation below


Before Change
template (value) NR_SpCellConfig_Type cs_NR_SS_SpCellConfig(template (omit) NR_ServingCellIndex_Type  p_NR_ServingCellIndex,

                                                              template (omit) CellGroupId               p_CellGroupId,

                                                              template (omit) MAC_CellGroupConfig       p_MAC_CellGroupConfig,

                                                              template (omit) PhysicalCellGroupConfig   p_PhysicalCellGroupConfig) :=

  { /* @status    APPROVED (ENDC, NR5GC, NR5GC_IRAT) */

    ServingCellIndex := p_NR_ServingCellIndex,

    CellGroupConfig := {

      CellGroupId := p_CellGroupId,

      SCellList   := {},

      MAC_CellGroupConfig:= { R15 := p_MAC_CellGroupConfig},

      PhysicalCellGroupConfig := { R15 := p_PhysicalCellGroupConfig}
    }

  };

After Change

template (value) NR_SpCellConfig_Type cs_NR_SS_SpCellConfig(template (omit) NR_ServingCellIndex_Type  p_NR_ServingCellIndex,

                                                              template (omit) CellGroupId               p_CellGroupId,

                                                              template (omit) NR_ASN1_MAC_CellGroupConfig_Type       p_MAC_CellGroupConfig,

                                                              template (omit) NR_ASN1_PhysicalCellGroupConfig_Type   p_PhysicalCellGroupConfig) :=

  { /* @status    APPROVED (ENDC, NR5GC, NR5GC_IRAT) */

    ServingCellIndex := p_NR_ServingCellIndex,

    CellGroupConfig := {

      CellGroupId := p_CellGroupId,

      SCellList   := {},

      MAC_CellGroupConfig := p_MAC_CellGroupConfig,

      PhysicalCellGroupConfig := p_PhysicalCellGroupConfig
    }

  };
MCC160 proposed implementation
  template (value) NR_SpCellConfig_Type cs_NR_SS_SpCellConfig(template (omit) NR_ServingCellIndex_Type p_NR_ServingCellIndex,

                                                              template (omit) CellGroupId             p_CellGroupId,

                                                              template (omit) MAC_CellGroupConfig     p_MAC_CellGroupConfig,

                                                              template (omit) PhysicalCellGroupConfig  p_PhysicalCellGroupConfig ) :=

  { /* @status    APPROVED (ENDC, NR5GC, NR5GC_IRAT) */

    ServingCellIndex := p_NR_ServingCellIndex,

    CellGroupConfig := {

      CellGroupId := p_CellGroupId,

      SCellList   := {},

      MAC_CellGroupConfig := fl_NR_Set_SpCellConfigMAC_CellGroupConfig(p_MAC_CellGroupConfig),

      PhysicalCellGroupConfig := fl_NR_Set_SpCellConfigPhysicalCellGroupConfig(p_PhysicalCellGroupConfig)
    }

  };

//------------------------------------------------------------------------------------------------------------------------

  /*

   * @desc      Set NR_ASN1_MAC_CellGroupConfig_Type with the given parameters MAC_CellGroupConfig

   */

  function fl_NR_Set_SpCellConfigMAC_CellGroupConfig(template(omit) MAC_CellGroupConfig p_MAC_CellGroupConfig) return template(omit) NR_ASN1_MAC_CellGroupConfig_Type

  { //@sic R5s201518 sic@

    var template(omit) NR_ASN1_MAC_CellGroupConfig_Type v_MAC_CellGroupConfig := omit;

    if (isvalue (p_MAC_CellGroupConfig)) {

        v_MAC_CellGroupConfig := { R15 := p_MAC_CellGroupConfig};

    }   

    return v_MAC_CellGroupConfig;

  }

//------------------------------------------------------------------------------------------------------------------------  /*

   * @desc      Set NR_ASN1_PhysicalCellGroupConfig_Type with the given parameters PhysicalCellGroupConfig   

   */

  function fl_NR_Set_SpCellConfigPhysicalCellGroupConfig(template(omit) PhysicalCellGroupConfig    p_PhysicalCellGroupConfig) return template(omit) NR_ASN1_PhysicalCellGroupConfig_Type

  { //@sic R5s201518 sic@

    var template(omit) NR_ASN1_PhysicalCellGroupConfig_Type v_PhysicalCellGroupConfig := omit;

    if (isvalue (p_PhysicalCellGroupConfig)) {

        v_PhysicalCellGroupConfig := { R15 := p_PhysicalCellGroupConfig};

    }   

    return v_PhysicalCellGroupConfig;

  }
2.2 Change 2
	Function name
	New function f_NR_SS_SpCellConfig()

	Reason for change
	As change 1

	Summary of change
	Adding wrapper function f_NR_SS_SpCellConfig to handle MAC_CellGroupConfig and PhysicalCellGroupConfig with both values and “omit”

	TTCN module
	

	MCC160 Comment
	Not needed – see change 1


New Function
function f_NR_SS_SpCellConfig(template (omit) NR_ServingCellIndex_Type  p_NR_ServingCellIndex,

                                 template (omit) CellGroupId               p_CellGroupId,

                                 template (omit) MAC_CellGroupConfig       p_MAC_CellGroupConfig,

                                 template (omit) PhysicalCellGroupConfig   p_PhysicalCellGroupConfig) return template (value) NR_SpCellConfig_Type

  {

    var NR_ASN1_MAC_CellGroupConfig_Type v_MAC_CellGroupConfig := omit;

    var NR_ASN1_PhysicalCellGroupConfig_Type v_PhysicalCellGroupConfig := omit;

    if (isvalue (p_MAC_CellGroupConfig))

    {

      v_MAC_CellGroupConfig := { R15 := p_MAC_CellGroupConfig};

    }

    if (isvalue (p_PhysicalCellGroupConfig))

    {

      v_PhysicalCellGroupConfig := { R15 := p_PhysicalCellGroupConfig};

    }

    return cs_NR_SS_SpCellConfig(p_NR_ServingCellIndex, p_CellGroupId, v_MAC_CellGroupConfig, v_PhysicalCellGroupConfig);

  }
2.3 Change 3
	Function name
	e.g. f_NR_ENDC_ReConfigAM_UM

	Reason for change
	As change 1

	Summary of change
	Replace use of template cs_NR_SS_SpCellConfig with call to new function f_NR_SS_SpCellConfig()
Note : Similar changes are needed in functions f_NR_ServingCellConfig_Init(),fl_NR_BuildRRCReconfigExistingDRBs(),f_TC_7_1_1_3_2b_ENDC_NR(),fl_TC_8_2_2_1_1_ENDC_NR_TestBody(), and f_TC_7_1_1_3_2b_NR5GC()

	TTCN module
	e.g. NR_CommonFunctions.ttcn

	MCC160 Comment
	Not needed – see change 1


Before Change
function f_NR_ENDC_ReConfigAM_UM(NR_CellId_Type p_CellId,

                                   integer p_NoOfNR_AMBearersN,

                                   integer p_NoOfNR_UMBearersM,

                                   template (omit) NR_RadioBearerList_Type p_SS_Drb_ConfigList := omit,

                                   template (omit) NR_RLC_BearerToAddModList_Type p_RLC_BearerConfigList := omit,

                                   template (omit) SRB_ToAddModList p_SRB_ToAddModList := omit,

                                   template (omit) DRB_ToAddModList p_DRB_ToAddModList := omit,

                                   template (omit) SecurityConfig p_SecurityConfig := omit,

                                   template (omit) MAC_CellGroupConfig p_MAC_CellGroupConfig := omit,

                                   NR_RRC_STATE_Type p_RRC_State := RRC_IDLE,

                                   template (omit) PDSCH_ServingCellConfig.nrofHARQ_ProcessesForPDSCH  p_NrofHARQ_ProcessesForPDSCH := omit

                                   ) runs on NR_BASE_PTC // @sic R5s201010 sic@

  {

.....................
    f_NR_SS_CommonRadioBearerConfig(p_CellId, v_SS_Drb_ConfigList);

    v_RadioBearerConfig := cs_38508_RadioBearerConfigDef(p_SRB_ToAddModList, v_DRB_ToAddModList, -, v_SecurityConfig);

    v_SCGConfig := f_NR_GetSCGConfigDef(p_CellId);

    if (isvalue(p_MAC_CellGroupConfig)) {

      v_SCGConfig.mac_CellGroupConfig := p_MAC_CellGroupConfig;

      f_NR_SS_ServingCellConfig(p_CellId, cs_TimingInfo_Now, cs_NR_ServingCellConfig_SpCell(cs_NR_SS_SpCellConfig(omit, omit, p_MAC_CellGroupConfig, omit)));

    }

    if (isvalue(p_NrofHARQ_ProcessesForPDSCH)) {

      v_SCGConfig.spCellConfig.spCellConfigDedicated.pdsch_ServingCellConfig := cs_NR_Pdsch_ServingCellConfigSetup(cs_NR_PDSCH_ServingCellConfigDef(-, -, p_NrofHARQ_ProcessesForPDSCH));

      f_NR_SS_CommonCellConfig(p_CellId, cads_NR_PDSCH_Config_REQ(p_CellId, cs_TimingInfo_Now, cs_NR_PDSCH_CellLevelConfigDef(v_SCGConfig.spCellConfig.spCellConfigDedicated.pdsch_ServingCellConfig.setup)));

    }

    v_SCGConfig.rlc_BearerToAddModList := v_RLC_BearerConfigList;

    f_NR_SendRRCReconfigurationContentsToEUTRA(v_RadioBearerConfig, v_SCGConfig, -, -, false, reconfigurationWithSync_CheckRACH);  //@sic R5s190541, R5s200647 sic@

    // Get Security info from EUTRA, calculate latest keys and configure SS

    // as only need to do this for the first reconfiguration @sic R5s200647, R5s201010 sic@

    if (p_RRC_State != RRC_CONNECTED)

    {

      f_ENDC_GetKeysActivateSecurity(p_CellId, EUTRA);

    }

    //RRCReconfigurationComplete not yet received from E-UTRA - Wait for it @sic R5s200647 sic@

    EUTRA.receive(cmr_EUTRA_NR_OctetData ("Complete")) -> value v_Coordination_MSG_Value;

    f_NR_CheckReceived_RRCReconfigurationComplete(valueof(v_Coordination_MSG_Value));

  }
After Change

function f_NR_ENDC_ReConfigAM_UM(NR_CellId_Type p_CellId,

                                   integer p_NoOfNR_AMBearersN,

                                   integer p_NoOfNR_UMBearersM,

                                   template (omit) NR_RadioBearerList_Type p_SS_Drb_ConfigList := omit,

                                   template (omit) NR_RLC_BearerToAddModList_Type p_RLC_BearerConfigList := omit,

                                   template (omit) SRB_ToAddModList p_SRB_ToAddModList := omit,

                                   template (omit) DRB_ToAddModList p_DRB_ToAddModList := omit,

                                   template (omit) SecurityConfig p_SecurityConfig := omit,

                                   template (omit) MAC_CellGroupConfig p_MAC_CellGroupConfig := omit,

                                   NR_RRC_STATE_Type p_RRC_State := RRC_IDLE,

                                   template (omit) PDSCH_ServingCellConfig.nrofHARQ_ProcessesForPDSCH  p_NrofHARQ_ProcessesForPDSCH := omit

                                   ) runs on NR_BASE_PTC // @sic R5s201010 sic@

  {

.............
    f_NR_SS_CommonRadioBearerConfig(p_CellId, v_SS_Drb_ConfigList);

    v_RadioBearerConfig := cs_38508_RadioBearerConfigDef(p_SRB_ToAddModList, v_DRB_ToAddModList, -, v_SecurityConfig);

    v_SCGConfig := f_NR_GetSCGConfigDef(p_CellId);

    if (isvalue(p_MAC_CellGroupConfig)) {

      v_SCGConfig.mac_CellGroupConfig := p_MAC_CellGroupConfig;

      f_NR_SS_ServingCellConfig(p_CellId, cs_TimingInfo_Now, cs_NR_ServingCellConfig_SpCell(f_NR_SS_SpCellConfig(omit, omit, p_MAC_CellGroupConfig, omit)));

    }

    if (isvalue(p_NrofHARQ_ProcessesForPDSCH)) {

      v_SCGConfig.spCellConfig.spCellConfigDedicated.pdsch_ServingCellConfig := cs_NR_Pdsch_ServingCellConfigSetup(cs_NR_PDSCH_ServingCellConfigDef(-, -, p_NrofHARQ_ProcessesForPDSCH));

      f_NR_SS_CommonCellConfig(p_CellId, cads_NR_PDSCH_Config_REQ(p_CellId, cs_TimingInfo_Now, cs_NR_PDSCH_CellLevelConfigDef(v_SCGConfig.spCellConfig.spCellConfigDedicated.pdsch_ServingCellConfig.setup)));

    }

    v_SCGConfig.rlc_BearerToAddModList := v_RLC_BearerConfigList;

    f_NR_SendRRCReconfigurationContentsToEUTRA(v_RadioBearerConfig, v_SCGConfig, -, -, false, reconfigurationWithSync_CheckRACH);  //@sic R5s190541, R5s200647 sic@

    // Get Security info from EUTRA, calculate latest keys and configure SS

    // as only need to do this for the first reconfiguration @sic R5s200647, R5s201010 sic@

    if (p_RRC_State != RRC_CONNECTED)

    {

      f_ENDC_GetKeysActivateSecurity(p_CellId, EUTRA);

    }

    //RRCReconfigurationComplete not yet received from E-UTRA - Wait for it @sic R5s200647 sic@

    EUTRA.receive(cmr_EUTRA_NR_OctetData ("Complete")) -> value v_Coordination_MSG_Value;

    f_NR_CheckReceived_RRCReconfigurationComplete(valueof(v_Coordination_MSG_Value));

  }
